

alpha — Compute interitem correlations (covariances) and Cronbach's alpha

[Description](#)[Quick start](#)[Menu](#)[Syntax](#)[Options](#)[Remarks and examples](#)[Stored results](#)[Methods and formulas](#)[Acknowledgment](#)[References](#)[Also see](#)

Description

`alpha` computes the interitem correlations or covariances for all pairs of variables in *varlist* and Cronbach's α statistic for the scale formed from them. At least two variables must be specified with `alpha`.

Quick start

Cronbach's alpha and the average interitem covariance of 6 variables

```
alpha v1 v2 v3 v4 v5 v6
```

As above, but standardize the variables to give average interitem correlation

```
alpha v1 v2 v3 v4 v5 v6, std
```

As above, and show table of item-test and item-rest correlations and the effect of removing each variable from the scale

```
alpha v1 v2 v3 v4 v5 v6, std item
```

As above, and also list each interitem correlation

```
alpha v1 v2 v3 v4 v5 v6, std item detail
```

As above, but force `v2` and `v5` to enter with reversed sign

```
alpha v1 v2 v3 v4 v5 v6, std item detail reverse(v2 v5)
```

As above, and generate `scalevar` with the summative score (scale)

```
alpha v1 v2 v3 v4 v5 v6, std item detail reverse(v2 v5) ///
generate(scalevar)
```

Menu

Statistics > Multivariate analysis > Cronbach's alpha

Syntax

```
alpha varlist [if] [in] [, options]
```

| <i>options</i> | Description |
|--------------------------------|--|
| asis | take sign of each item as is |
| casewise | delete cases with missing values |
| detail | list individual interitem correlations and covariances |
| generate(<i>newvar</i>) | save the generated scale in <i>newvar</i> |
| item | display item-test and item-rest correlations |
| label | include variable labels in output table |
| min(#) | must have at least # observations for inclusion |
| reverse(<i>varlist</i>) | reverse signs of these variables |
| std | standardize items in the scale to mean 0, variance 1 |

by and collect are allowed; see [\[U\] 11.1.10 Prefix commands](#).

Options

Options

asis specifies that the sense (sign) of each item be taken as presented in the data. The default is to determine the sense empirically and reverse the scorings for any that enter negatively.

casewise specifies that cases with missing values be deleted listwise. The default is pairwise computation of covariances and correlations.

detail lists the individual interitem correlations and covariances.

generate(*newvar*) specifies that the scale constructed from *varlist* be saved in *newvar*. Unless **asis** is specified, the sense of items entering negatively is automatically reversed. If **std** is also specified, the scale is constructed by using standardized (mean 0, variance 1) values of the individual items. Unlike most Stata commands, **generate()** does not use casewise deletion. A score is created for every observation for which there is a response to at least one item (one variable in *varlist* is not missing). The summative score is divided by the number of items over which the sum is calculated.

item specifies that item-test and item-rest correlations and the effects of removing an item from the scale be displayed. **item** is valid only when more than two variables are specified in *varlist*.

label requests that the detailed output table be displayed in a compact format that enables the inclusion of variable labels.

min(#) specifies that only cases with at least # observations be included in the computations. **casewise** is a shorthand for **min(*k*)**, where *k* is the number of variables in *varlist*.

reverse(*varlist*) specifies that the signs (directions) of the variables (items) in *varlist* be reversed. Any variables specified in **reverse()** that are not also included in **alpha**'s *varlist* are ignored.

std specifies that the items in the scale be standardized (mean 0, variance 1) before summing.

Remarks and examples

stata.com

Cronbach's alpha (Cronbach 1951) assesses the reliability of a summative rating (Likert 1932) scale composed of the variables (called *items*) specified. The set of items is often called a *test* or *battery*. A scale is simply the sum of the individual item scores, reversing the scoring for statements that have negative correlations with the factor (for example, attitude) being measured. Scales can be formed by using the raw item scores or standardized item scores.

The reliability α is defined as the square of the correlation between the measured scale and the underlying factor. If you think of a test as being composed of a random sample of items from a hypothetical domain of items designed to measure the same thing, α represents the expected correlation of one test with an alternative form containing the same number of items. The square root of α is the estimated correlation of a test with errorless true scores (Nunnally and Bernstein 1994, 235).

In addition to reporting α , alpha generates the summative scale from the items (variables) specified and automatically reverses the sense of any when necessary. Stata's decision can be overridden by specifying the `reverse(varlist)` option.

Because it concerns reliability in measuring an unobserved factor, α is related to factor analysis. The test should be designed to measure one factor, and, because the scale will be composed of an unweighted sum, the factor loadings should all contribute roughly equal information to the score. Both of these assumptions can be verified with `factor`; see [MV] `factor`. Equality of factor loadings can also be assessed by using the `item` option.

► Example 1

To illustrate alpha, we apply it, first without and then with the `item` option, to the automobile dataset after randomly introducing missing values:

```
. use https://www.stata-press.com/data/r17/automiss
(1978 automobile data)

. alpha price headroom rep78 trunk weight length turn displ, std
Test scale = mean(standardized items)
Reversed item:  rep78

Average interitem correlation:      0.5251
Number of items in the scale:      8
Scale reliability coefficient:      0.8984
```

The scale derived from our somewhat arbitrarily chosen automobile items (variables) appears to be reasonable because the estimated correlation between it and the underlying factor it measures is $\sqrt{0.8984} \approx 0.9478$ and the estimated correlation between this battery of eight items and all other eight-item batteries from the same domain is 0.8984. Because the “items” are not on the same scale, it is important that `std` was specified so that the scale and its reliability were based on the sum of standardized variables. We could obtain the scale in a new variable called `sc` with the `gen(sc)` option.

Though the scale appears reasonable, we include the `item` option to determine if all the items fit the scale:

```
. alpha price headroom rep78 trunk weight length turn displ, std item
Test scale = mean(standardized items)
```

| Item | Obs | Sign | Item-test correlation | Item-rest correlation | Average interitem correlation | alpha |
|--------------|-----|------|--------------------------|--------------------------|-------------------------------------|--------|
| price | 70 | + | 0.5260 | 0.3719 | 0.5993 | 0.9128 |
| headroom | 66 | + | 0.6716 | 0.5497 | 0.5542 | 0.8969 |
| rep78 | 61 | - | 0.4874 | 0.3398 | 0.6040 | 0.9143 |
| trunk | 69 | + | 0.7979 | 0.7144 | 0.5159 | 0.8818 |
| weight | 64 | + | 0.9404 | 0.9096 | 0.4747 | 0.8635 |
| length | 69 | + | 0.9382 | 0.9076 | 0.4725 | 0.8625 |
| turn | 66 | + | 0.8678 | 0.8071 | 0.4948 | 0.8727 |
| displacement | 63 | + | 0.8992 | 0.8496 | 0.4852 | 0.8684 |
| Test scale | | | | | 0.5251 | 0.8984 |

“Test” denotes the additive scale; here 0.5251 is the average interitem correlation, and 0.8984 is the alpha coefficient for a test scale based on all items.

“Obs” shows the number of nonmissing values of the items; “Sign” indicates the direction in which an item variable entered the scale; “-” denotes that the item was reversed. The remaining four columns in the table provide information on the effect of one item on the scale.

Column four gives the item-test correlations. Apart from the sign of the correlation for items that entered the scale in reversed order, these correlations are the same numbers as those computed by the commands

```
. alpha price headroom rep78 trunk weight length turn displ, std gen(sc)
. pcorr sc price headroom rep78 trunk weight length turn displ
```

Typically, the item-test correlations should be roughly the same for all items. Item-test correlations may not be adequate to detect items that fit poorly because the poorly fitting items may distort the scale. Accordingly, it may be more useful to consider item-rest correlations (Nunnally and Bernstein 1994), that is, the correlation between an item and the scale that is formed by all other items. The average interitem correlations (covariances if `std` is omitted) of all items, excluding one, are shown in column six. Finally, column seven gives Cronbach's α for the test scale, which consists of all but the one item.

Here neither the `price` item nor the `rep78` item seems to fit well in the scale in all respects. The item-test and item-rest correlations of `price` and `rep78` are much lower than those of the other items. The average interitem correlation increases substantially by removing either `price` or `rep78`; apparently, they do not correlate strongly with the other items. Finally, we see that Cronbach's α coefficient will increase from 0.8984 to 0.9128 if the `price` item is dropped, and it will increase from 0.8984 to 0.9143 if `rep78` is dropped. For well-fitting items, we would of course expect that α decreases by shortening the test.

▶ Example 2

The variable names for the automobile data are reasonably informative. This may not always be true; items in batteries commonly used to measure personality traits, attitudes, values, etc., are usually named with indexed names, for example, `item12a`, `item12b`, `item12c`. The `label` option forces `alpha` to produce the same statistical information in a more compact format that leaves room to include variable (item) labels. In this compact format, `alpha` excludes the number of nonmissing values of the items, displays the statistics using fewer digits, and uses somewhat cryptic headers:

```
. alpha price headroom rep78 trunk weight length turn displ, std item label detail
```

```
Test scale = mean(standardized items)
```

| Items | S | it-cor | ir-cor | ii-cor | alpha | Label |
|--------------|---|--------|--------|--------|-------|--------------------------|
| price | + | 0.526 | 0.372 | 0.599 | 0.913 | Price |
| headroom | + | 0.672 | 0.550 | 0.554 | 0.897 | Headroom (in.) |
| rep78 | - | 0.487 | 0.340 | 0.604 | 0.914 | Repair record 1978 |
| trunk | + | 0.798 | 0.714 | 0.516 | 0.882 | Trunk space (cu. ft.) |
| weight | + | 0.940 | 0.910 | 0.475 | 0.863 | Weight (lbs.) |
| length | + | 0.938 | 0.908 | 0.473 | 0.862 | Length (in.) |
| turn | + | 0.868 | 0.807 | 0.495 | 0.873 | Turn circle (ft.) |
| displacement | + | 0.899 | 0.850 | 0.485 | 0.868 | Displacement (cu. in.) |
| Test scale | | | | 0.525 | 0.898 | mean(standardized items) |

```
Interitem correlations (reverse applied) (obs=pairwise, see below)
```

| | | | | |
|--------------|---------|----------|--------|--------------|
| | price | headroom | rep78 | trunk |
| price | 1.0000 | | | |
| headroom | 0.1174 | 1.0000 | | |
| rep78 | -0.0479 | 0.1955 | 1.0000 | |
| trunk | 0.2748 | 0.6841 | 0.2777 | 1.0000 |
| weight | 0.5093 | 0.5464 | 0.3624 | 0.6486 |
| length | 0.4511 | 0.5823 | 0.3162 | 0.7404 |
| turn | 0.3528 | 0.4067 | 0.4715 | 0.5900 |
| displacement | 0.5537 | 0.5166 | 0.3391 | 0.6471 |
| | weight | length | turn | displacement |
| weight | 1.0000 | | | |
| length | 0.9425 | 1.0000 | | |
| turn | 0.8712 | 0.8589 | 1.0000 | |
| displacement | 0.8753 | 0.8422 | 0.7723 | 1.0000 |

```
Pairwise number of observations
```

| | | | | |
|--------------|--------|----------|-------|--------------|
| | price | headroom | rep78 | trunk |
| price | 70 | | | |
| headroom | 62 | 66 | | |
| rep78 | 59 | 54 | 61 | |
| trunk | 65 | 61 | 59 | 69 |
| weight | 60 | 56 | 52 | 60 |
| length | 66 | 61 | 58 | 64 |
| turn | 62 | 58 | 56 | 62 |
| displacement | 59 | 58 | 51 | 58 |
| | weight | length | turn | displacement |
| weight | 64 | | | |
| length | 60 | 69 | | |
| turn | 57 | 61 | 66 | |
| displacement | 54 | 58 | 56 | 63 |

Because the `detail` option was also specified, the interitem correlation matrix was printed, together with the number of observations used for each entry (because these varied across the matrix). Note the negative sign attached to `rep78` in the output, indicating the sense in which it entered the scale.

Better-looking output with less-cryptic headers is produced if the linesize is set to a value of at least 100:

```
. set linesize 100
. alpha price headroom rep78 trunk weight length turn displ, std item label
Test scale = mean(standardized items)
```

| Item | Obs | Sign | Item-test | Item-rest | Interitem | alpha | Label |
|--------------|-----|------|-----------|-----------|-----------|--------|--------------------------|
| | | | corr. | corr. | corr. | | |
| price | 70 | + | 0.5260 | 0.3719 | 0.5993 | 0.9128 | Price |
| headroom | 66 | + | 0.6716 | 0.5497 | 0.5542 | 0.8969 | Headroom (in.) |
| rep78 | 61 | - | 0.4874 | 0.3398 | 0.6040 | 0.9143 | Repair record 1978 |
| trunk | 69 | + | 0.7979 | 0.7144 | 0.5159 | 0.8818 | Trunk space (cu. ft.) |
| weight | 64 | + | 0.9404 | 0.9096 | 0.4747 | 0.8635 | Weight (lbs.) |
| length | 69 | + | 0.9382 | 0.9076 | 0.4725 | 0.8625 | Length (in.) |
| turn | 66 | + | 0.8678 | 0.8071 | 0.4948 | 0.8727 | Turn circle (ft.) |
| displacement | 63 | + | 0.8992 | 0.8496 | 0.4852 | 0.8684 | Displacement (cu. in.) |
| Test scale | | | | | 0.5251 | 0.8984 | mean(standardized items) |

◀

Users of `alpha` require some standard for judging values of α . We paraphrase [Nunnally and Bernstein \(1994, 265\)](#): In the early stages of research, modest reliability of 0.70 or higher will suffice; values in excess of 0.80 often waste time and funds. In contrast, where measurements on individuals are of interest, a reliability of 0.80 may not be nearly high enough. Even with a reliability of 0.90, the standard error of measurement is almost one-third as large as the standard deviation of test scores; a reliability of 0.90 is the minimum that should be tolerated, and a reliability of 0.95 should be considered the desirable standard.

Stored results

`alpha` stores the following in `r()`:

Scalars

| | |
|-----------------------|--|
| <code>r(alpha)</code> | scale reliability coefficient |
| <code>r(k)</code> | number of items in the scale |
| <code>r(cov)</code> | average interitem covariance |
| <code>r(rho)</code> | average interitem correlation if <code>std</code> is specified |

Matrices

| | |
|-----------------------------------|--|
| <code>r(Alpha)</code> | scale reliability coefficient |
| <code>r(ItemTestCorr)</code> | item-test correlation |
| <code>r(ItemRestCorr)</code> | item-rest correlation |
| <code>r(MeanInterItemCov)</code> | average interitem covariance |
| <code>r(MeanInterItemCorr)</code> | average interitem correlation if <code>std</code> is specified |

If the `item` option is specified, results are stored as row matrices for the `k` subscales when one variable is removed.

Methods and formulas

Let x_i , $i = 1, \dots, k$, be the variables over which α is to be calculated. Let s_i be the sign with which x_i enters the scale. If `asis` is specified, $s_i = 1$ for all i . Otherwise, principal-factor analysis is performed on x_i , and the first factor's score is predicted; see [\[MV\] factor](#). s_i is -1 if correlation of the x_i and the predicted score is negative and $+1$ otherwise.

Let r_{ij} be the correlation between x_i and x_j , c_{ij} be the covariance, and n_{ij} be the number of observations used in calculating the correlation or covariance. The average correlation is

$$\bar{r} = \frac{\sum_{i=2}^k \sum_{j=1}^{i-1} s_i s_j n_{ij} r_{ij}}{\sum_{i=2}^k \sum_{j=1}^{i-1} n_{ij}}$$

and the average covariance similarly is

$$\bar{c} = \frac{\sum_{i=2}^k \sum_{j=1}^{i-1} s_i s_j n_{ij} c_{ij}}{\sum_{i=2}^k \sum_{j=1}^{i-1} n_{ij}}$$

Let c_{ii} denote the variance of x_i , and define the average variance as

$$\bar{v} = \frac{\sum_{i=1}^k n_{ii} c_{ii}}{\sum_{i=1}^k n_{ii}}$$

If `std` is specified, the scale reliability α is calculated as defined by the general form of the Spearman–Brown Prophecy Formula (Nunnally and Bernstein 1994, 232; Allen and Yen 1979, 85–88):

$$\alpha = \frac{k\bar{r}}{1 + (k - 1)\bar{r}}$$

This expression corresponds to α under the assumption that the summative rating is the sum of the standardized variables (Nunnally and Bernstein 1994, 234). If `std` is not specified, α is defined (Nunnally and Bernstein 1994, 232 and 234) as

$$\alpha = \frac{k\bar{c}}{\bar{v} + (k - 1)\bar{c}}$$

Let x_{ij} reflect the value of item i in the j th observation. If `std` is specified, the j th value of the scale computed from the k x_{ij} items is

$$S_j = \frac{1}{k_j} \sum_{i=1}^k s_i S(x_{ij})$$

where $S()$ is the function that returns the standardized (mean 0, variance 1) value if x_{ij} is not missing and returns zero if x_{ij} is missing. k_j is the number of nonmissing values in x_{ij} , $i = 1, \dots, k$. If `std` is not specified, $S()$ is the function that returns x_{ij} or returns missing if x_{ij} is missing.

Lee Joseph Cronbach (1916–2001) was born in Fresno, California. He participated in a study of gifted children and completed high school at the age of 14. He obtained a doctoral degree in educational psychology from the University of Chicago, where he later served as an assistant professor. During World War II, he applied his expertise as a military psychologist for the U.S. Navy.

Cronbach is widely known for his paper on the alpha coefficient, which measures test reliability. While his work on the reliability coefficient was focused on psychological testing and education, Cronbach's alpha is used in several fields. He wrote an article with Paul Meehl on psychological test validity and later published a book with Goldine Gleser introducing generalizability theory.

Cronbach made a lasting impact on program evaluations through his collaborative research with faculty at Stanford, along with his book *Designing Evaluations of Educational and Social Programs*. He also played an active role in the American Psychological Association's work on test standards and later became president of the American Psychological Association. His many contributions are reflected in the honors he received from the American Education Research Association, the Educational Testing Service, and other associations.

Acknowledgment

This improved version of `alpha` was written by Jeroen Weesie of the Department of Sociology at Utrecht University, The Netherlands.

References

- Acock, A. C. 2018. *A Gentle Introduction to Stata*. 6th ed. College Station, TX: Stata Press.
- Allen, M. J., and W. M. Yen. 1979. *Introduction to Measurement Theory*. Monterey, CA: Brooks/Cole.
- Cronbach, L. J. 1951. Coefficient alpha and the internal structure of tests. *Psychometrika* 16: 297–334. <https://doi.org/10.1007/BF02310555>.
- Likert, R. A. 1932. A technique for the measurement of attitudes. *Archives of Psychology* 140: 5–55.
- Nunnally, J. C., and I. H. Bernstein. 1994. *Psychometric Theory*. 3rd ed. New York: McGraw–Hill.
- Shavelson, R. J., and G. Gleser. 2002. Lee J. Cronbach (1916–2001). *American Psychologist* 57: 360–361.
- Tarlov, A. R., J. E. Ware, Jr., S. Greenfield, E. C. Nelson, E. Perrin, and M. Zubkoff. 1989. The medical outcomes study. An application of methods for monitoring the results of medical care. *Journal of the American Medical Association* 262: 925–930. <https://doi.org/10.1001/jama.1989.03430070073033>.

Also see

[MV] **factor** — Factor analysis

[R] **icc** — Intraclass correlation coefficients